ECONOMICS

Introduction

In the ECONOMICS sub-program there are three basic commands used for the calculation of energy costs; they are UTILITY-RATE, BLOCK-CHARGE, and SCHEDULE.

- The UTILITY-RATE command describes the most basic features of a tariff: units, uniform energy and demand costs, monthly charges, minimum charges, etc.
- The BLOCK-CHARGE command is used to define energy or demand charges that vary according to the amount used.
- The SCHEDULE command is used to coordinate the operation of UTILITY-RATEs and BLOCK-CHARGEs.

Example ECONOMICS Input

The best way to learn to input tariffs in ECONOMICS is to study the input examples in conjunction with the following command and keyword descriptions. A simple example is given below. See also Appendix B and Examples 1 through 5 at the end of this section.

```
INPUT   ECONOMICS     ..

ELEC-TARIFF = UTILITY-RATE
              RESOURCE        = ELECTRICITY
              BLOCK-CHARGES = (INVBLK)    ..

INVBLK      = BLOCK-CHARGE
              BLOCK1-TYPE     = ENERGY
              BLOCK1-DATA     = (500,  .0535
                                 400,  .0725
                                   1,  .1245   ..

GAS-TARIFF  = UTILITY-RATE
              RESOURCE        = NATURAL-GAS
              MONTH-CHGS      = (15.)
              ENERGY-COST     = .75    ..

ECONOMICS-REPORT      SUMMARY = (ES-D)    ..

END   ..

COMPUTE ECONOMICS     ..
```

Description of ECONOMICS Input Instructions

This section contains descriptions of the input instructions required to run the energy cost calculations in ECONOMICS at a basic level. Additional commands and keywords are listed in the *Supplement (2.1E).* (The ECONOMICS program can also be used to do life-cycle cost analysis; see the *Reference Manual (2.1A).*)

UTILITY–RATE

This command contains the most basic features of a tariff: units, uniform energy and demand cost rates, monthly charges, minimum charges, etc. One UTILITY–RATE command is entered for each type of energy or fuel used in the previous PLANT run. UTILITY–RATE also allows costs to be defined for energy that varies by time, but not by quantity. For charges that vary by quantity, such as blocks of energy with different costs, this command references one or more BLOCK–CHARGE commands. When energy is billed in a time of use format a UTILITY–RATE keyword accepts the u-name of a schedule which defines these charges for the entire year. Time of use demand charges are more complicated and must be defined using the BLOCK–CHARGE command.

On an hourly basis, each UTILITY–RATE defined in ECONOMICS sums the energy into total and peak values for each billing period. If a time of use energy charge schedule has been defined, the hourly values are multiplied by the scheduled value and summed for the billing period. The UTILITY–RATE also passes the hourly data to the associated BLOCK–CHARGEs. During each billing cycle, the UTILITY–RATE adjusts the metered energy and demand values for any minimum quantities required. These adjusted billing quantities are then used as the basis for the energy cost calculations. Block costs computed by any associated BLOCK–CHARGEs are then added. The costs are then adjusted by any minimum monthly charges and/or rate limitations.

u-name — required so that each rate can be listed in the summary reports. Up to 15 different UTILITY–RATEs may be defined.

RESOURCE — is a required keyword that determines which fuel or energy type is being valued. The code-words associated with this keyword are the same as in PLANT: STEAM, CHILLED–WATER, ELECTRICITY, NATURAL–GAS, LPG, DIESEL–OIL, FUEL–OIL, COAL, METHANOL, or OTHER–FUEL. When OTHER–FUEL is referenced, the OTHER–FUEL–NAME as defined in PLANT will be used in all reports as the RESOURCE.

MONTH–CHGS — accepts a list in parentheses of up to 12 numeric values that add a fixed monthly charge to each billing cycle. Many utilities refer to this as a customer charge. This value can range from 0.0 to 100,000 $/month and defaults to 0.0. If a single value is entered, the value will be used for all 12 billing periods. If fewer than 12 are entered, the last value entered will be used for all remaining billing periods. This is true of all numeric lists in ECONOMICS.

ENERGY–CHG

accepts a numeric value that allows you to specify an energy charge that is constant with time and quantity. The units are $/UNIT and can range from $0.0 to $100,000,000/UNIT. In the absence of any other charges, this keyword will default in accordance with Table 5.1. The UNIT value is the unit as defined or defaulted in the ENERGY–RESOURCE command of PLANT.

Table 5.1

Default ENERGY–CHG Values for UTILITY–RATE

RESOURCE	VALUE	$/UNIT
CHILLED–WATER	12,000 Btu/ton	0.12
COAL	24,580,000 Btu/ton	30.00
DIESEL–OIL	138,700 Btu/gal	1.05
ELECTRICITY	3,412.97 Btu/kWh	0.07
FUEL–OIL	138,700 Btu/gal	1.19
LPG	95,500 Btu/gal	0.97
METHANOL	63,500 Btu/gal	1.13
NATURAL–GAS	100,000 Btu/therm	0.50
OTHER–FUEL	1,000,000 Btu/unit	0.95
STEAM	1,000,000 Btu/unit	13.00

ENERGY–CHG–SCH

accepts the u-name of a SCHEDULE which specifies an ENERGY–CHG that varies by time of day, week and/or season. The units in the schedule should be $/UNIT. This schedule is used for all time of use energy billing (demand time of use billing is more complex, and requires the use of multiple BLOCK–CHARGEs). If both an ENERGY–CHG and ENERGY–CHG–SCH are defined, the values will add.

DEMAND–CHGS

accepts a list in parentheses of 12 values that allows you to specify a demand charge that is constant with quantity but may vary by billing period. The units are $/peak-UNIT and can range from $0.0 to $100,000,000/peak-UNIT. The default is 0.0. As with other lists, as few as one value may be entered in the list, and the last value will be used for the remaining billing periods.

BLOCK–CHARGE

accepts a list in parentheses of up to 10 u-names of BLOCK–CHARGEs. These BLOCK–CHARGEs can be used to calculate either energy or demand charges that vary according to quantity. In addition, time of use demand charges are calculated using the BLOCK–CHARGE format.

MIN–MON–CHGS

accepts a list in parentheses of 12 values that place a floor on the cost of a fuel or utility for each billing period in which costs are calculated. This value can range from $0 to $100,000,000 per month and defaults to $0. As with other lists, not all 12 values need be entered; the last value entered will be used for all remaining billing periods. Note that the minimum charge excludes any customer charge, taxes, surcharges, or energy cost adjustments.

MIN–MON–DEM–CHGS

accepts a list in parentheses of up to 12 values that specify a variable minimum monthly charge calculated on the basis of billing demand. The value entered has units of $/kW, and the default is 0. Some utilities do not charge directly for demand, but embed demand charges in kWh/kW rate structures. This keyword allows demand charges to be levied in the event actual demand is exceptionally high relative to total energy use. The total minimum month charges will be the sum of the constant and variable minimum charges as specified by MIN–MON–CHGs and MIN–MON–DEM–CHGs.

RATE–LIMITATION

accepts a numeric value in dollars per unit that places a ceiling on the maximum effective rate that will be assessed on a utility or fuel for any month. This value can range from $0.0 to $100,000,000/UNIT and defaults to $100,000,000/UNIT. The RATE–LIMITATION excludes MONTH–CHGs and charges arising from ENERGY–COST–ADJustments. In addition, the RATE–LIMITATION cannot cause the total bill to drop below the fixed MIN–MON–CHGs plus the MIN–MON–DEM–CHGs.

BILLING–DAYS

accepts a list in parentheses of up to 12 values. If you input less than 12, the last value entered will be the default for all unentered values. The default is 31, or the last day of the month. All costs, etc. reported in output reports will be based on the billing-day. For example, if the billing-day is 17, energy usage for each month listed in the reports will be from the 18th day of the previous month, through the 17th day of the current month. You may elect to explicitly enter the billing day for each month when it is desired to closely match existing utility bills. This may be important if the meter is not regularly read on the same day each month.

BLOCK–CHARGE

BLOCK–CHARGEs are used to calculate energy or demand costs that vary according to the amount consumed. BLOCK–CHARGEs are also used for time of use demand charges. Up to 30 BLOCK–CHARGEs may be defined, and up to 10 may be referenced by each UTILITY–RATE. The same BLOCK–CHARGE may be referenced by more than one UTILITY–RATE; the program automatically makes as many working copies as are required.

u-name — is a unique user-defined name that must be entered to identify this command.

BLOCK–SCH — accepts the u-name of a SCHEDULE which defines the period over which energy or demand specified in the parent UTILITY–RATE is accumulated. For all energy charges, the schedule is used only for seasonal changes in block charges, such as winter vs. summer rates; time of use rates should be defined directly in the UTILITY–RATE via the ENERGY–CHG–SCH. It is not necessary for seasons to begin and end on the BILLING–DAYS; the program will prorate charges when a seasonal change occurs in the middle of the billing period. You should be careful, however, to ensure that one or more other BLOCK–CHARGEs are defined for the periods in which this BLOCK–CHARGE is inactive.

This schedule may be used to define as many seasons as may be required. For conventional block demand charges, seasonal changes are handled identically to energy block charges. For demand charges levied on a time of use basis, such as peak, shoulder, and off-peak, this schedule may vary on an hourly and daily basis, as well as seasonal basis.

SCH–FLAG — specifies the flag value in the BLOCK–SCH that indicates when this BLOCK–CHARGE is active. The default is 1.0

BLOCK1–TYPE — accepts a code-word used to define the type of block calculation that will follow.

ENERGY — the default; used when the cost calculations will be done on a per unit basis of energy consumption.

KWH/KW — is used to specify calculations for energy consumption where each block size is defined as a multiplier on demand. When the number of kWh that can be in a given kWh/kW block is limited to a maximum value, a limit can be specified (see BLOCK–DATA).

KWH/KW–LIMITSUM — is used when the sum of the kWh in a series of ENERGY blocks is limited by a kWh/kW value. The maximum value is specified in the BLOCK–DATA.

In the left column above, bold-faced words are commands, non-bold words are keywords, and italicized words are code-words.

DEMAND

specifies that this set of BLOCK–CHARGEs will be used for calculating demand charges.

BLOCK1–DATA

accepts a list enclosed by parentheses of up to 10 sets of data. For ENERGY and DEMAND blocks, each set consists of two entries in the order (block-size, cost/unit). For KWH/KW and KWH/KW–LIMITSUM types, each set consists of three entries in the order (block-size, cost/unit, limit).

The first entry of each set indicates the size of the block to which the cost/unit will be applied. Blocks are increments; hence each successive BLOCK–DATA entry covers the next size block and its cost. Rates written as "up to X" must be translated. The range is from 0.0 to 100,000,000 and there is no default.

The second entry of each set indicates the cost/unit to be applied against the energy or demand falling within this block.

For KWH/KW and KWH/KW–LIMITSUM types, the third entry is the limit. An entry of 0.0 means there is no limit.

1. When the BLOCK–UNIT is KWH/KW, the limit has units of kWh and is the maximum quantity of energy that can be charged in this block.
2. When the BLOCK–UNIT is KWH/KW–LIMITSUM, this entry has units of kWh/kW and, when multiplied by demand, is the maximum sum of all energy that can be charged in this block and all previous blocks. Usually, one limit will apply to a series of blocks. In this case, the same limit should be entered for each of the affected blocks.

Example: A utility charges $0.05 for the first 5,000 kWh, $0.04 for the next 10,000 kWh, and $0.035 for the remainder. Input is as follows:

```
BLOCK1–TYPE    =ENERGY
               $kWh,  COST
BLOCK1–DATA    =( 5000, .05,      $ SET #1
                  10000, .04,     $ SET #2
                      1, .035)    $ SET #3
```

Assuming that the BLOCK1–DATA is not followed by an entry for BLOCK2–DATA, all remaining energy will go into the last block; therefore, its size does not matter. Note also that, while this format is easy to read, it is not mandatory. All data could have been entered on a single line.

BLOCK2–TYPE

When energy calculations are being made, this keyword allows the type of energy calculation to change. For example, a utility may start with a series of ENERGY blocks and then change to kWh/kW blocks. Alternatively, energy blocks may

switch to demand.

BLOCK2–DATA

This keyword is used in an identical fashion to BLOCK1–DATA. When BLOCK2–TYPE is the same as BLOCK1–TYPE, this keyword allows an additional 10 sets of data to be entered. If the previous example also contained kWh/kW blocks, input might be as follows:

```
BLOCK1-TYPE =    ENERGY
                 $kWh          COST
BLOCK1-DATA =    ( 5000,       .05,      $ SET #1
                   10000,      .04,      $ SET #2
                   30000,      .035)     $ SET #3
BLOCK2-TYPE =    KWH/KW
                 $kWh/kW       COST      MAX kWh
BLOCK2-DATA =    (100,         .03,        40000,
                  200,         .02,            0,
                    1,         .015,           0 )
```

Note that data for kWh/kW is entered in sets of three, with the third entry being the limit. The limit must be specified, even if it is zero (no limit).

BLOCK3–TYPE

Same

BLOCK3–DATA

Same

TOU–SEASON–LINKS

This keyword is used only with a time of use rate where different BLOCK–CHARGEs are used at different times of the day, and is required only when seasonal changes in rates occur on a day which does not coincide with the billing-day. In this case, the two BLOCK–CHARGEs overlap in the same billing period and must be linked so that the correct energy and/or demand charges can be determined for both blocks. (Charges for each block must be based on the same use period, such as on-peak, and then prorated based on the number of hours each block was active during the billing period. See the section on Yearly, Seasonal, and Time of Use BLOCK–CHARGEs for more information).

Input is a list of u-name(s) in parenthesis of the linked BLOCK–CHARGE(s) which share the same billing period. Both BLOCK–CHARGEs must reference each other via this keyword. If only two seasons are used, this BLOCK–CHARGE will overlap with only one other BLOCK–CHARGE, so that only one u-name is entered. If more than two seasons are used, such as winter, spring, summer and fall, this BLOCK–CHARGE will overlap with two other BLOCK–CHARGEs. For example, a BLOCK–CHARGE

representing spring will overlap with both winter and summer BLOCK–CHARGEs. In this case, the u-names of the winter and summer BLOCK–CHARGEs are input.

If a UTILITY–RATE has a block structure for both energy and demand charges, the same BLOCK–CHARGE may be used to model both. When modeling both, the energy and demand BLOCK–TYPE keywords may be specified in any order. Alternatively, separate BLOCK–CHARGEs can be used for energy and demand. This may be useful when the BLOCK–SCH for the energy and demand charges do not coincide. As previously described, each UTILITY–RATE can reference up to 10 BLOCK–CHARGEs.

Yearly, Seasonal, and Time of Use BLOCK-CHARGEs

BLOCK-CHARGEs can be used to model yearly, seasonal, or time of use (i.e., time of day) rates:

Example of a BLOCK-CHARGE Yearly Rate

A yearly rate is very straightforward to model. One or more BLOCK-CHARGEs are defined without defining a BLOCK-SCH. The rate will then be used all year. For example, the following input models a yearly rate:

```
ELEC-TARIFF = UTILITY-RATE
              RESOURCE          = ELECTRICITY
              BLOCK-CHARGES     = (INVBLK)   ..

INVBLK      = BLOCK-CHARGE
              BLOCK1-TYPE       = ENERGY
                                  $SIZE   COST
              BLOCK1-DATA       = (500,   .0535,
                                   400,   .0725,
                                     1,   .1245)   ..
```

Example of a BLOCK-CHARGE Seasonal Rate

A seasonal rate is also straightforward. As before, one or more BLOCK-CHARGEs are defined, and a BLOCK-SCH is also defined to indicate when each BLOCK-CHARGE is actively used. The following is an example of a seasonal rate:

```
ELEC-TARIFF = UTILITY-RATE
              RESOURCE          = ELECTRICITY
              BILLING-DAYS      = (31)
              BLOCK-CHARGES     = (WINTER-BLK,
                                   SUMMER-BLK) ..

WINTER-BLK  = BLOCK-CHARGE
              BLOCK-SCH         = SEASONS-SCH
              SCH-FLAG          = 1
              BLOCK1-TYPE       = ENERGY
                                  $SIZE   COST
              BLOCK1-DATA       = (1000,  .07,
                                      1,  .10 )   ..

SUMMER-BLK  = BLOCK-CHARGE
              BLOCK-SCH         = SEASONS-SCH
              SCH-FLAG          = 2
              BLOCK1-TYPE       = ENERGY
                                  $SIZE   COST
              BLOCK1-DATA       = (500,   .06,
                                     1,   .09)  ..
```

```
SEASONS-SCH = SCHEDULE    THRU MAY 15   (ALL)   (1,24)  (1)
                          THRU SEP 15   (ALL)   (1,24)  (2)
                          THRU DEC 31   (ALL)   (1,24)  (1)  ..
```

Note that in this example, the season changes from winter to summer on May 15, but the billing day is at the end of the month. This means that, during the month of May, the winter block-charge is used in the first half of the month, and the summer block-charge is used for the latter half. In this case, DOE-2 prorates the costs between the two block-charges in the same way that most utilities do:

a. Costs for each BLOCK–CHARGE are computed using all of the energy consumed during the entire billing period. For example, the cost for the WINTER–BLOCK is computed using the energy billed for the entire month of May, not just the energy metered for the first half of the month.

b. The costs are then prorated based on the number of hours each BLOCK–CHARGE was active. For example, the cost computed for WINTER–BLOCK is multiplied by 15/31 days.

The above example is for an energy type BLOCK–CHARGE
(BLOCK1–TYPE = ENERGY, KWH/KW, or KWH/KW–LIMITSUM).
DEMAND type blocks are handled similarly:

a. The cost for each DEMAND type BLOCK–CHARGE is computed using the maximum demand found in the entire billing period.

b. The costs are then prorated based on the number of hours each BLOCK–CHARGE was active.

Both DOE-2 and the utility companies prorate charges based on the number of days rather than on actual energy consumed because the standard utility meter accumulates a single value of energy and a single value of demand for the billing period; information on the distribution of energy and demand usage is not available. Report ES-F shows how the proration is done. For each BLOCK–CHARGE, the "metered energy" is the energy metered during the period defined by the BLOCK–SCH. The "billing energy" is the energy metered during the entire billing period (i.e, the month). The billing energy is the amount used to compute the cost. Usually, the metered energy and the billing energy will be the same except when the season changes in the middle of a billing period. In this case, the "prorate factor" is used to adjust the actual charges. Logically, the prorate factors of two seasonal BLOCK–CHARGEs sharing the same billing period will add up to 1.0